
1   /*
2   Script: MuseScore_Hotkeys_v1.1.ahk 
3   Author: "Wyvern". 
4   Date: 3rd May 2017.
5   
6   This Script is designed to enhance the Rhythm Entry mode available in MuseScore 2.1.
7   It allows the left-hand to select Note/Rest/Dot/Tuplet modes with the Control keys,
8   while the right-hand simultaneously enters a duration using the Number Pad. It is a
9   less sophisticated version of the Denemo approach.
10   
11   The Script requires AutoHotkeys Version 1.1.24.05 or later. The software is Open 
12   Source, and may be downloaded from the AutoHotkey Foundation under License GNU GPLv2. 
13   SciTE4AutoHotkey is highly recommended for editing and debugging the Script.
14   A compiled *.exe version can also be created that runs without AutoHotkey installed.
15   
16   The Directive "#IfWinActive ahk_exe MuseScore.exe" ensures that the Hotkeys are only 
17   enabled while the MuseScore Window is Active.
18   
19   You are welcome to use, modify, or otherwise adapt the Script as you see fit. 
20   However, this is entirely at your own risk. Please use the MuseScore forum to provide
21   "Wyvern" with helpful feedback and suggestions for improving the Script.
22   
23   The Number Pad 'Modifiers' are shown below.
24   
25   NumPad (1-9) : Note
26   Win+NumPad (1-9) : Rest
27   Alt+NumPad (1-9) : Dotted Note
28   Win+Alt+NumPad (1-9) : Dotted Rest
29   Ctrl+NumPad (2-9) : Tuplet (acts on the previous duration entered)
30   
31   Ctrl+Alt+H : Help Box
32   Ctrl+Alt+S : Suspend/Resume Hotkeys (allows normal MuseScore action)
33   Ctrl+Alt+X : Exit
34   
35   ; -------------------------------------------------------------------------------------
36   ; Scratch Pad 
37   ; -------------------------------------------------------------------------------------
38   ; To edit the Script, comment-out the "#IfWinActive..." lines. It is highly recommended 
39   ; that you use a Scratch Pad to avoid overwriting your code with 'Send' characters!
40   ; Just place your mouse cursor in the Scratch Pad area before triggering any Hotkeys.
41   ;
42   ;
43   ; 
44   ;
45   ;
46   ;
47   ;
48   ;
49   */
50   
51   ; -------------------------------------------------------------------------------------
52   ; Directives
53   ; -------------------------------------------------------------------------------------
54   
55   #NoEnv ; Recommended for compatibility with new AutoHotkey releases.
56   #SingleInstance Force ; Replaces script (Reloads).
57   #Warn ; Enables warnings to assist with detecting common errors.
58   
59   ; -------------------------------------------------------------------------------------
60   ; Initialisation
61   ; -------------------------------------------------------------------------------------
62   
63   SendMode, Event                 ; Allows a delay to be set between keystrokes.
64   SetKeyDelay, 10 ; The (default) delay in milliseconds.
65   SetWorkingDir %A_ScriptDir% ; Ensures a consistent starting directory.      
66   SetNumLockState, On ; In case we forget!
67   Suspend, On ; Suspends Hotkeys until after the initial Help Box.
68   
69   ; -------------------------------------------------------------------------------------

70   ; Global Variables
71   ; -------------------------------------------------------------------------------------
72   
73   Buffer := 0 ; The current key character.
74   
75   ; -------------------------------------------------------------------------------------
76   ; Main Script
77   ; -------------------------------------------------------------------------------------
78   
79   if A_IsSuspended ; This is true at start-up.
80   {
81   gosub, Help      ; Displays the initial Help Box.
82   Suspend, Off ; Enables the Hotkeys.
83   }
84   return
85   
86   ; -------------------------------------------------------------------------------------
87   ; Always active Hotkeys
88   ; -------------------------------------------------------------------------------------
89   
90   ; Help Hotkey
91   ^!h::
92   Suspend, Permit
93   gosub, Help
94   return
95   
96   ; Suspend/Resume Hotkey
97   ^!s::
98   Suspend
99   if A_IsSuspended
100   MsgBox, 0, MuseScore HotKeys. Version 1.1.
101   , Hotkeys are suspended.`n`nCtrl+Alt+S to Resume.
102   , 5 ; Five second timeout.
103   else
104   MsgBox, 0, MuseScore HotKeys. Version 1.1.
105   , Hotkeys are active!`n`nCtrl+Alt+S to Suspend.
106   , 5 ; Five second timeout.
107   return
108   
109   ; Exit Application Hotkey
110   ^!x::
111   Suspend
112   ExitApp
113   
114   ;-------------------------------------------------------------------------------------
115   
116   #IfWinActive ahk_exe MuseScore.exe ; Enables Hotkeys when MuseScore Window is Active.   
117   
118   ;-------------------------------------------------------------------------------------
119   ; Control Keys (Modifiers)
120   ; -------------------------------------------------------------------------------------
121   
122   ; Disables the Ctrl keys acting alone.
123   ~LCtrl Up::return
124   ~RCtrl Up::return
125   
126   ; Disables the Win keys acting alone.
127   ~LWin Up::return
128   ~RWin Up::return
129   
130   ; Disables the Alt key acting alone.
131   ~Alt Up::return
132   
133   ; -------------------------------------------------------------------------------------
134   ; Note Hotkeys
135   ; -------------------------------------------------------------------------------------
136   
137   ; Number Pad : No Modifier key.
138   ;Blocked.

139   Numpad0::
140   NumpadDot::
141   return
142   ; Active 
143   Numpad1::
144   Numpad2::
145   Numpad3::
146   Numpad4::
147   Numpad5::
148   Numpad6::
149   Numpad7::
150   Numpad8::
151   Numpad9::
152   ; Retrieves the key value.
153   StringTrimLeft, Buffer, A_ThisHotkey, 6
154   ; Sends the MuseScore string
155   Send, %Buffer%
156   return
157   
158   ; -------------------------------------------------------------------------------------
159   ; Rest Hotkeys
160   ; -------------------------------------------------------------------------------------
161   
162   ; Number Pad : With Win key (#) pressed.
163   ; Blocked.
164   #Numpad0::
165   #NumpadDot::
166   return
167   ; Active 
168   #Numpad1::
169   #Numpad2::
170   #Numpad3::
171   #Numpad4::
172   #Numpad5::
173   #Numpad6::
174   #Numpad7::
175   #Numpad8::
176   #Numpad9::
177   ; Retrieves the key value.
178   StringTrimLeft, Buffer, A_ThisHotkey, 7
179   ; Sends the MuseScore string
180   Send, 0%Buffer%0 ; '0' is the MS Rest key.
181   return
182   
183   ; -------------------------------------------------------------------------------------
184   ; Dotted Note Hotkeys
185   ; -------------------------------------------------------------------------------------
186   
187   ; Number Pad : With Alt key (!) pressed.
188   ; Blocked.
189   !Numpad0::
190   !NumpadDot::
191   return
192   ; Active 
193   !Numpad1::
194   !Numpad2::
195   !Numpad3::
196   !Numpad4::
197   !Numpad5::
198   !Numpad6::
199   !Numpad7::
200   !Numpad8::
201   !Numpad9::
202   ; Retrieves the key value.
203   StringTrimLeft, Buffer, A_ThisHotkey, 7
204   ; Sends the MuseScore string
205   Send, .%Buffer%. ; '.' is the MS Dot key   
206   return
207   

208   ; -------------------------------------------------------------------------------------
209   ; Dotted Rest Hotkeys
210   ; -------------------------------------------------------------------------------------
211   
212   ; Number Pad : With Win and Alt keys (#!) pressed.
213   ; Blocked.
214   #!Numpad0::
215   #!NumpadDot::
216   return
217   ; Active   
218   #!Numpad1::
219   #!Numpad2::
220   #!Numpad3::
221   #!Numpad4::
222   #!Numpad5::
223   #!Numpad6::
224   #!Numpad7::
225   #!Numpad8::
226   #!Numpad9::
227   ; Retrieves the key value.
228   StringTrimLeft, Buffer, A_ThisHotkey, 8
229   ; Sends the MuseScore string
230   Send, 0.%Buffer%0. ; '0' and '.' are the MS Rest and Dot characters, respectively.   
231   return
232   
233   ; -------------------------------------------------------------------------------------
234   ; Tuplet Hotkeys
235   ; -------------------------------------------------------------------------------------
236   
237   ; Number Pad : With Ctrl key (^) pressed.
238   ; Blocked.
239   ^Numpad0::
240   ^Numpad1::
241   ^NumpadDot::
242   return
243   ; Active   
244   ^Numpad2::
245   ^Numpad3::
246   ^Numpad4::
247   ^Numpad5::
248   ^Numpad6::
249   ^Numpad7::
250   ^Numpad8::
251   ^Numpad9::
252   ; Retrieves the key value.
253   StringTrimLeft, Buffer, A_ThisHotkey, 7
254   ; Sends the MuseScore string, shifting left to use the previous key duration.
255   Send, {Left}^%Buffer% ; Cursor left, Ctrl+Tuplet.   
256   return
257   
258   #IfWinActive ; Turns off the initial Directive. Just to be tidy!
259   
260   ; -------------------------------------------------------------------------------------
261   ; Subroutines
262   ; -------------------------------------------------------------------------------------
263   
264   ; Help Box:  Press Ctrl+Alt+H to display.
265   Help:
266   MsgBox, 0, MuseScore HotKeys. Version 1.1.,
267   (
268   This AutoHotkey Script should be used with the
269   MuseScore Rhythm Entry mode. The Number Pad
270   enters Note or Rest durations. The Win and Alt
271   keys modify the Number Pad, as specified in
272   the list below. They are pressed 'on the fly' 
273   while entering the durations (Tuplets excepted).
274   The Tuplets act on the prior duration entered.
275   
276   NumPad (1-9) : Note

277   Win+NumPad (1-9) : Rest
278   Alt+NumPad (1-9) : Dotted Note
279   Win+Alt+NumPad (1-9) : Dotted Rest
280   Ctrl+NumPad (2-9) : Tuplet
281   
282   Ctrl+Alt+H : This Help Box
283   Ctrl+Alt+S : Suspend/Resume Hotkeys
284   Ctrl+Alt+X : Exit
285   )
286   IfWinNotActive ahk_exe nightly.exe
287   MsgBox, 0, MuseScore HotKeys. Version 1.1.
288   , The MuseScore Window is not Active!
289   , 5 ; Five second timeout.   
290   return
291   

